

WEINTEK LABS., INC.

Unix Timestamp (Epoch Time) Converter

Converts dates into Unix/Epoch
timestamps on HMI

Demo Project

Contents

- 1. Overview and Operation 1
 - Overview..... 1
 - Operation 1
 - Calculation Explained 1
 - Macro details..... 2
- 2. Setting up the Screen 3
- 3. Addresses 4

1. Overview and Operation

Overview

Unix Time (also known as Epoch Time) is defined as the number of seconds that have elapsed since 00:00:00 Coordinated Universal Time (UTC), 1st of January, 1970. This demo project can convert a date into Unix Time.

Operation

Enter the Year, Month, Day, Hour, Minute, and Second in the corresponding fields. The project will automatically do the conversion and display the result in Unix/Epoch Time field. To convert the HMI RTC time, tap [Local Time] button. The RTC time will be automatically filled in.



The screenshot shows a web-based interface titled "EPOCH Converter". It features six input fields for time components: Year (2016), Month (6), Day (6), Hour (15), Minute (27), and Second (22). To the right of these fields is a button labeled "Local Time". Below the input fields, a large downward-pointing arrow indicates the conversion process. Underneath the arrow, the text "Unix/Epoch Time" is displayed above a large output field containing the value "1465226842".

Calculation Explained

POSIX:2008 section 4.15 defines "seconds since the Epoch" as an abstract measure approximating the number of seconds that have elapsed since the Epoch, excluding leap seconds. The formula given is

```
tm_sec + tm_min*60 + tm_hour*3600 + tm_yday*86400 +  
(tm_year-70)*31536000 + ((tm_year-69)/4)*86400 -  
((tm_year-1)/100)*86400 + ((tm_year+299)/400)*86400
```

This calculation assumes that leap years occur in each year that is either divisible by 4 but not divisible by 100, or is divisible by 400.

Macro details


The macro used in this project is listed below:

```
macro_command main()  
  
short yy, mm, dd, hh, min, ss  
unsigned short md[13] = {0,31,59,90,120,151,181,212,243,273,304,334}  
unsigned short md_leap[13] = {0,31,60,91,121,152,182,213,244,274,305,335}  
unsigned int epoch = 0  
unsigned int a,b,c,d,e,f  
  
GetData(yy, "Local HMI", LW, 100, 1)  
GetData(mm, "Local HMI", LW, 101, 1)  
GetData(dd, "Local HMI", LW, 102, 1)  
GetData(hh, "Local HMI", LW, 103, 1)  
GetData(min, "Local HMI", LW, 104, 1)  
GetData(ss, "Local HMI", LW, 105, 1)  
  
yy = yy - 1900  
mm = mm-1  
if yy%4==0 then  
    dd = md_leap[mm] + dd - 1  
else  
    dd = md[mm] + dd - 1  
end if  
  
a = (yy-70) * 31536000  
b = (yy-69) / 4 * 86400  
c = (yy-1) /100 * 86400  
d = (yy + 299) / 400 * 86400  
  
epoch = ss + min*60 + hh*3600 + dd*86400 + a + b - c + d  
SetData(epoch, "Local HMI", LW, 0, 1)  
  
end macro_command
```

2. Setting up the Screen

Step 1. Create 6 Numeric (input) objects for entering a date or RTC time, and 1 Numeric (display) object for showing Unix time.

Step 2. Build a macro that converts RTC time into Unix time.



3. Addresses

The addresses of key objects used in this demonstration are listed below.

Object	Address	Object ID	Description
Window 10			
Numeric	LW-100	NE_1	Year field.
Numeric	LW-101	NE_0	Month field.
Numeric	LW-102	NE_2	Day field.
Numeric	LW-103	NE_3	Hour field.
Numeric	LW-104	NE_4	Minute field.
Numeric	LW-105	NE_5	Second field.
Numeric	LW-0	ND_1	Shows the Unix timestamp.
Function Key		FK_1	Fills in the date using HMI RTC time.